
    Changelog

All notable changes to this project will be documented in this file.
0.7.1 - 2025-10-12
Added
	Zoi.to_json_schema/1 support for metadata (e.g., example, description)
	guides/quickstart_guide.md added to the documentation

0.7.0 - 2025-10-10
Added
	Zoi.to_json_schema/1 function to convert Zoi schemas to JSON Schema format

Changed
	Zoi.array/2 fixed path in errors when parsing arrays
	Zoi.regex/2 fixed regex compile errors when used in module attributes

0.6.6 - 2025-10-08
Added
	Zoi.metadata/1 - option to add metadata to types for documentation purposes

Changed
	Zoi.example/1 deprecated in favor of Zoi.metadata/1

0.6.5 - 2025-10-07
Added
	Zoi.example/1 option to add example values to types for documentation and testing purposes

0.6.4 - 2025-09-30
Added
	Zoi.downcase/1 refinement to validate if a string is in lowercase
	Zoi.upcase/1 refinement to validate if a string is in uppercase
	Zoi.hex/1 refinement to validate if a string is a valid hexadecimal

0.6.3 - 2025-09-27
Added
	keys in Zoi.object/2 data structure
	Zoi.struct/2 type to parse structs and maps into structs
	Zoi.Struct module with helper functions to work with structs. This module offers two main functions:	Zoi.Struct.enforce_keys/1: List of keys that must be present in the struct
	Zoi.Struct.struct_keys/1: List of keys and their default values to be used with defstruct



0.6.2 - 2025-09-26
Added
	Zoi.literal/2 type to accept only a specific literal value

Changed
	Refactor all errors to be generated on type creation instead of parsing time

0.6.1 - 2025-09-08
Added
	Zoi.null/1 type to accept only nil values
	Zoi.positive/1 refinement for numbers to accept only positive values
	Zoi.negative/1 refinement for numbers to accept only negative values

0.6.0 - 2025-09-07
Added
	Zoi.required/2 type to enforce presence of a value in keyword and object types

Changed
	Zoi.object/2 now uses mfa to call inner transform function
	Zoi.keyword/2 have all fields set as optional by default, use Zoi.required/2 to enforce presence of a value

0.5.7 - 2025-09-06
Changed
	Zoi.parse!/3 Error message

0.5.6 - 2025-09-05
Added
	Zoi.parse!/3 function that raises an error if parsing fails
	Zoi.type_spec/2 function that returns the Elixir type spec for a given Zoi schema, implemented for all types

0.5.5 - 2025-09-03
Added
	Zoi.keyword/2 type

Changed
	Zoi.struct/2 now works with the new Zoi.keyword/2 type
	Improved Zoi.transform/2 documentation

0.5.4 - 2025-08-29
Added
	Guide for converting keys from maps
	Guide for generating schema from JSON structure

0.5.3 - 2025-08-29
Changed
	Fix transform and refinement types

0.5.2 - 2025-08-28
Added
	Zoi.prettify_errors/2 added docs
	Zoi.extend/3 type

Changed
	Zoi.map/3 now parses key and value types correctly
	Fix encapsulated types ignoring refinements and transforms when parsing

0.5.1 - 2025-08-17
Changed
	Zoi.prettify_errors/1 don't return \n at the end of the string anymore

0.5.0 - 2025-08-17
Added
	Zoi.atom/1 type
	Zoi.string_boolean/1 type
	Zoi.union/2 custom error messages
	Zoi.intersection/2 custom error messages
	Zoi.to_struct/2 transform

Changed
	Zoi.boolean/1 does not coerce values besides "true" and "false" anymore. For coercion of other values, use Zoi.string_boolean/1 type.

0.4.0 - 2025-08-14
Added
	Zoi.Context module to provide context when parsing data

Changed
	Zoi.object/2 will not automatically parse objects with inputs that differ from the string/atom keys map format. For example:

schema = Zoi.object(%{
  name: Zoi.string(),
  age: Zoi.integer()
})
Zoi.object(schema, %{"name" => "John", "age" => 30})
{:error, _errors}
To make this API work, you can pass coerce: true option to Zoi.object/2. This will make the object parser to check from the Map input if the keys are strings or atoms and fetch it's values automatically.
schema = Zoi.object(%{
  name: Zoi.string(),
  age: Zoi.integer()
})
Zoi.object(schema, %{"name" => "John", "age" => 30}, coerce: true)
{:ok, %{name: "John", age: 30}}
0.3.4 - 2025-08-09
Added
	Zoi.min/2, Zoi.max/2, Zoi.gt/2, Zoi.gte/2, Zoi.lt/2, Zoi.lte/2 refinements for Zoi.time/1 type
	Zoi.min/2, Zoi.max/2, Zoi.gt/2, Zoi.gte/2, Zoi.lt/2, Zoi.lte/2 refinements for Zoi.date/1 type
	Zoi.min/2, Zoi.max/2, Zoi.gt/2, Zoi.gte/2, Zoi.lt/2, Zoi.lte/2 refinements for Zoi.datetime/1 type
	Zoi.min/2, Zoi.max/2, Zoi.gt/2, Zoi.gte/2, Zoi.lt/2, Zoi.lte/2 refinements for Zoi.naive_datetime/1 type

0.3.3 - 2025-08-09
Added
	Zoi.time/1 type
	Zoi.date/1 type
	Zoi.datetime/1 type
	Zoi.naive_datetime/1 type

0.3.2 - 2025-08-09
Added
	Zoi.decimal/1 type
	Zoi.min/2, Zoi.max/2, Zoi.gt/2, Zoi.gte/2, Zoi.lt/2, Zoi.lte/2 refinements for Zoi.decimal/1 type

0.3.1 - 2025-08-08
Added
	Zoi.ISO.time/1 type
	Zoi.ISO.date/1 type
	Zoi.ISO.datetime/1 type
	Zoi.ISO.to_time_struct/1 transform
	Zoi.ISO.to_date_struct/1 transform
	Zoi.ISO.to_datetime_struct/1 transform
	Zoi.ISO.to_naive_datetime/1 transform
	Zoi.prettify_errors/1 function to format errors in a human-readable way

0.3.0 - 2025-08-07
Added
	Zoi.email/0 format
	Zoi.url/0 format
	Zoi.uuid/1 format

Changed
	Removed Zoi.email/1, now use Zoi.email/0 that will automatically use the Zoi.string/1 type
	All refinements now accept a :message option to customize the error message

0.2.3 - 2025-08-06
Added
	Zoi.map/3 type
	Zoi.intersection/2 type
	Zoi.gt/2 refinement
	Zoi.gte/2 refinement
	Zoi.lt/2 refinement
	Zoi.lte/2 refinement

0.2.2 - 2025-08-06
Added
	Guides for using Zoi in Phoenix controllers
	New Zoi.tuple/2 type
	New Zoi.any/1 type
	New Zoi.nullable/2 type

Changed
	Improved error messages for all validations and types
	Zoi.treefy_errors/1 now returns a more human-readable structure
	Zoi.optional/2 cannot accept nil as a value anymore. Use Zoi.nullable/2 instead.
	Zoi.optional/2 inside Zoi.object/2 now handles optional fields correctly

0.2.1 - 2025-08-06
Added
	Custom error messages for primitive types

Changed
	Zoi.number/2 now returns proper error message

0.2.0 - 2025-08-05
Added
	mfa to Zoi.refine/2 and Zoi.transform/2 functions
	accumulator errors to Zoi.refine/2 and Zoi.transform/2 functions
	Zoi.array/2 type
	Zoi.length/2, Zoi.min/2 and Zoi.max/2 validators for arrays

Changed
	errors are now returned as a list of %Zoi.Error{} structs

[image: Buy Me a Coffee at ko-fi.com]Zoi is a schema validation library for Elixir, designed to provide a simple and flexible way to define and validate data.
Installation
zoi to your list of dependencies in mix.exs:
def deps do
  [
    {:zoi, "~> 0.7"}
  ]
end
Usage
You can create schemas for various data types, including strings, integers, floats, booleans, arrays, maps, and more. Zoi supports a wide range of validation rules and transformations.
Parsing Data
Here's a simple example of how to use Zoi to validate a string:
# Define a schema with a string type
iex> schema = Zoi.string() |> Zoi.min(3)
iex> Zoi.parse(schema, "hello")
{:ok, "hello"}
iex> Zoi.parse(schema, "hi")
{:error, [%Zoi.Error{message: "too small: must have at least 3 characters"}]}


# Add transforms to a schema
iex> schema = Zoi.string() |> Zoi.trim()
iex> Zoi.parse(schema, "    world    ")
{:ok, "world"}
You can also validate structured maps:
# Validate a structured data in a map
iex> schema = Zoi.object(%{name: Zoi.string(), age: Zoi.integer(), email: Zoi.email()})
iex> Zoi.parse(schema, %{name: "John", age: 30, email: "john@email.com"})
{:ok, %{name: "John", age: 30, email: "john@email.com"}}
iex> Zoi.parse(schema, %{email: "invalid-email"})
{:error, [
    %Zoi.Error{path: [:name], message: "is required"},
    %Zoi.Error{path: [:age], message: "is required"},
    %Zoi.Error{path: [:email], message: "invalid email format"}
]}
and arrays:
# Validate an array of integers
iex> schema = Zoi.array(Zoi.integer() |> Zoi.min(0)) |> Zoi.min(2)
iex> Zoi.parse(schema, [1, 2, 3])
{:ok, [1, 2, 3]}
iex> Zoi.parse(schema, [1, "2"])
{:error, [%Zoi.Error{path: [1], message: "invalid type: must be an integer"}]}
And many more possibilities, including nested schemas, custom validations and data transformations. Check the official docs for more details.
Types
Zoi can infer types from schemas, allowing you to leverage Elixir's @type and @spec annotations for documentation
defmodule MyApp.Schema do
  @schema Zoi.string() |> Zoi.min(2) |> Zoi.max(100)
  @type t :: unquote(Zoi.type_spec(@schema))
end
This will generate the following type specification:
@type t :: binary()
This also applies to complex types, such as Zoi.object/2:
defmodule MyApp.User do
  @schema Zoi.object(%{
    name: Zoi.string() |> Zoi.min(2) |> Zoi.max(100),
    age: Zoi.integer() |> Zoi.optional(),
    email: Zoi.email()
  })
  @type t :: unquote(Zoi.type_spec(@schema))
end
Which will generate:
@type t :: %{
  required(:name) => binary(),
  optional(:age) => integer(),
  required(:email) => binary()
}
Errors
When validation fails, Zoi returns a list of errors, each containing a message and the path to the invalid data. Even when erros are nested, Zoi will return all errors in a flattened list.
iex> schema = Zoi.object(%{name: Zoi.string(), age: Zoi.integer()})
iex> Zoi.parse(schema, %{name: 123, age: "thirty"})
{:error, [
    %Zoi.Error{path: [:name], message: "invalid type: must be a string"},
    %Zoi.Error{path: [:age], message: "invalid type: must be an integer"}
]}
You can view the error in a map format using the Zoi.treefy_errors/1 function:
iex> Zoi.treefy_errors(errors)
%{
  name: ["invalid type: must be a string"],
  age: ["invalid type: must be an integer"]
}
You can also customize error messages:
iex> schema = Zoi.string(error: "not a string")
iex> Zoi.parse(schema, :hi)
{:error, [%Zoi.Error{message: "not a string"}]}
Metadata
You can attach metadata to schemas using the :metadata option. This metadata can be useful for documentation, testing, or other any purpose your application may require.
iex> schema = Zoi.string(metadata: [id: "1", description: "A simple string"])
iex> Zoi.metadata(schema)
[id: "1", description: "A simple string"]
You can use this feature to create self-documenting schemas, with example and tests. For example:
defmodule MyApp.UserSchema do
  @schema Zoi.object(
            %{
              name: Zoi.string() |> Zoi.min(2) |> Zoi.max(100),
              age: Zoi.integer() |> Zoi.optional()
            },
            metadata: [
              example: %{name: "Alice", age: 30},
              doc: "A user schema with name and optional age",
              moduledoc: "Schema representing a user with name and optional age"
            ]
          )

  @moduledoc """
  #{Zoi.metadata(@schema)[:moduledoc]}
  """

  @doc """
  #{Zoi.metadata(@schema)[:doc]}
  """
  def schema, do: @schema
end

defmodule MyApp.UserSchemaTest do
  use ExUnit.Case
  alias MyApp.UserSchema

  test "example matches schema" do
    example = Zoi.metadata(UserSchema.schema())[:example]
    assert {:ok, example} == Zoi.parse(UserSchema.schema(), example)
  end
end
Guides
Check the official guides for more examples and use cases:
	Quickstart Guide
	Using Zoi to generate OpenAPI specs
	Validating controller parameters
	Converting Keys From Object
	Generating Schemas from JSON

Acknowledgements
Zoi is inspired by Zod and Joi, providing a similar experience for Elixir.


  

    Quickstart Guide

This guide will help you get started with Zoi on your Elixir project.
Installation
Add zoi to your list of dependencies in mix.exs:
def deps do
  [
    {:zoi, "~> 0.7"}
  ]
end
Defining Schemas
You can define schemas using a variety of built-in types and validation rules. Here are some examples:
# String schema with minimum length
schema = Zoi.string() |> Zoi.min(3)
# Integer schema with range
schema = Zoi.integer() |> Zoi.min(1) |> Zoi.max(100)
# Email schema
schema = Zoi.email()
# Object schema with nested fields
schema = Zoi.object(%{
  name: Zoi.string() |> Zoi.regex(~r/^[a-zA-Z ]+$/),
  age: Zoi.integer() |> Zoi.min(0),
  email: Zoi.email()
})
# Array schema with item validation
schema = Zoi.array(Zoi.integer() |> Zoi.min(0)) |> Zoi.min(2)
Validating Data
You can validate data against your defined schemas using the Zoi.parse/2 function:
schema = Zoi.object(%{
  name: Zoi.string() |> Zoi.regex(~r/^[a-zA-Z ]+$/),
  age: Zoi.integer() |> Zoi.min(0),
  email: Zoi.email()
})

Zoi.parse(schema, %{name: "John Doe", age: 30, email: "john@email.com"})
# {:ok, %{name: "John Doe", age: 30, email: "john@email.com"}}

Zoi.parse(schema, %{name: "John123", age: -5, email: "invalid-email"})
# {:error, [
#   %Zoi.Error{path: [:name], message: "invalid string: must match a pattern ~r/^[a-zA-Z ]+$/"},
#   %Zoi.Error{path: [:age], message: "too small: must be at least 0"},
#   %Zoi.Error{path: [:email], message: "invalid email format"}
# ]}
Use Cases
You can use Zoi in various scenarios, such as:
	Validating user input in web forms
	Validating API request parameters
	Normalizing data before processing
	Generating OpenAPI specifications
	Integrating with external systems and validating responses
	And more!

Checkout our guides for more detailed examples and use cases.


  

    Using Zoi to generate OpenAPI specs

The OpenAPI Specification define a standard interface for HTTP APIs. The 3.1 version officially supports 100% compatibility with the latest draft (2020-12) of JSON Schema specification to describe the structure of request and response payloads.
Zoi implements the conversion between Zoi schemas and JSON Schema, which can be used to generate OpenAPI specs for your Phoenix application.
You can define a Zoi schema for a user resource:
schema = Zoi.object(%{
  id: Zoi.integer() |> Zoi.min(1),
  name: Zoi.string() |> Zoi.min(1) |> Zoi.max(100),
  age: Zoi.optional(Zoi.integer() |> Zoi.min(0))
})
Then, you can convert it to JSON Schema:
json_schema = Zoi.to_json_schema(schema)
This will generate the following structure:
%{
  "$schema" => "https://json-schema.org/draft/2020-12/schema",
  type: :object,
  properties: %{
    id: %{type: :integer, minimum: 1},
    name: %{type: :string, minLength: 1, maxLength: 100},
    age: %{type: :integer, minimum: 0}
  },
  required: [:id, :name],
  additionalProperties: false
}
You can then use this JSON Schema to define the request and response bodies in your OpenAPI specification.
Integrating with Phoenix
To integrate Zoi with Phoenix and generate OpenAPI specs, you can create a module that defines your API endpoints and their corresponding Zoi schemas. Then, you can use a library like Oaskit to generate the OpenAPI specification. Follow the Oaskit guide to do the initial setup, then you can use Zoi schemas in your endpoint definitions.
defmodule MyAppWeb.UserController do
  use MyAppWeb, :controller
  use Oaskit.Controller

  alias MyApp.Users

  @user_schema Zoi.object(%{
    id: Zoi.integer() |> Zoi.min(1),
    name: Zoi.string() |> Zoi.min(1) |> Zoi.max(100),
    age: Zoi.optional(Zoi.integer() |> Zoi.min(0))
  })

  @user_spec Zoi.to_json_schema(@user_schema)

  operation :create,
    summary: "Create User",
    request_body: {@user_spec [required: true]},
    responses: [ok: {@user_spec, []}]

    def create(conn, params) do
      ## Validate params using Zoi or Oaskit
    end
end
With this setup, you get both request validation and OpenAPI documentation generation out of the box.
For more details on how to validate parameters in a Phoenix controller using Zoi, see the Validating controller parameters guide.
You can also validate the request params using Oaskit directly, see the Oaskit documentation for more details.


  

    Validating controller parameters

One common use case for Zoi is validating request parameters in a Phoenix controller, before they reach your business logic or database layer.
Here's a typical controller setup using Ecto to validate the incoming params:
defmodule MyAppWeb.UserController do
  use MyAppWeb, :controller

  alias MyApp.Users

  def create(conn, params) do
    case Users.create_user(params) do
      {:ok, user} ->
        conn
        |> put_status(:created)
        |> render("show.json", user: user)

      {:error, changeset} ->
        conn
        |> put_status(:unprocessable_entity)
        |> render(MyAppWeb.ChangesetView, "error.json", changeset: changeset)
    end
  end
end
This works well when your API payload matches the database schema, but:
	You may want stricter or custom validation rules.
	Different field names than your schema.
	Your API shape may differ from your DB schema.
	You want to fail early, before calling your domain logic.
	Custom behaviour (optional fields, specific formats)

defmodule MyAppWeb.UserController do
  use MyAppWeb, :controller

  alias MyApp.Users

  @user_params Zoi.object(%{
    name: Zoi.string(),
    email: Zoi.email() |> Zoi.min(4) |> Zoi.max(100),
    age: Zoi.integer(coerce: true) |> Zoi.min(18) |> Zoi.max(100)
  }, coerce: true)

  def create(conn, params) do
    case Zoi.parse(@user_params, params) do
      {:ok, valid_params} ->
        case Users.create_user(valid_params) do
          {:ok, user} ->
            conn
            |> put_status(:created)
            |> render("show.json", user: user)

          {:error, changeset} ->
            conn
            |> put_status(:unprocessable_entity)
            |> render(MyAppWeb.ChangesetView, "error.json", changeset: changeset)
        end

      {:error, errors} ->
        conn
        |> put_status(:unprocessable_entity)
        |> render(MyAppWeb.ErrorView, "error.json", errors: Zoi.treefy_errors(errors))
    end
  end
end
	Zoi.parse/2 returns {:ok, data} or {:error, [Zoi.Error.t()]}.
	Zoi.treefy_errors/1 transforms flat error lists into a structured tree, useful for forms or APIs.

Validating query parameters
One powerful use case for Zoi is validating and normalizing query parameters passed to your Phoenix controller.
Imagine a paginated endpoint like this:
GET /api/posts?page=2&limit=50&sort=-published_at
possible validations:
	Ensure page and limit are integers
	Apply default values if not provided
	Validate sort against allowed fields

@query_schema Zoi.object(%{
  page: Zoi.default(Zoi.integer(coerce: true) |> Zoi.min(1), 1),
  limit: Zoi.default(Zoi.integer(coerce: true) |> Zoi.min(1) |> Zoi.max(100), 10),
  sort: Zoi.optional(Zoi.string() |> Zoi.enum(["published_at", "-published_at"]))
})
Use it on your controller:
def index(conn, params) do
  case Zoi.parse(@query_schema, params) do
    {:ok, query} ->
      posts = Blog.list_posts(query)
      render(conn, "index.json", posts: posts)

    {:error, errors} ->
      conn
      |> put_status(:unprocessable_entity)
      |> json(%{errors: Zoi.treefy_errors(errors)})
  end
end
And sending invalid params: GET /api/posts?page=0&limit=200&sort=name
will return a structured error response like this:
{
  "errors": {
    "page": ["too small: must be at least 1"],
    "limit": ["too big: must be at most 100"],
    "sort": ["Invalid option: must be one of published_at, -published_at"]
  }
}


  

    Converting Keys From Object

In Zoi you can also apply complex transformations to keys in maps. This is useful when you want to convert keys from one format to another, such as from camelCase to snake_case.
For example, consider the following JSON data:
{
  "firstName": "John",
  "lastName": "Doe",
  "address": {
    "streetAddress": "21 2nd Street",
    "city": "New York"
  }
}
If you want to transform the keys to snake_case, you can use Zoi.transform/2 as follows:
defmodule MyApp.User do
  @moduledoc false

  def schema() do
    Zoi.object(%{
      "firstName" => Zoi.string(),
      "lastName" => Zoi.string(),
      "address" =>
        Zoi.object(%{
          "streetAddress" => Zoi.string(),
          "city" => Zoi.string()
        })
        |> to_snake_case()
    })
    |> to_snake_case()
  end

  defp to_snake_case(schema) do
    schema
    |> Zoi.transform(fn map ->
      for {k, v} <- map, into: %{}, do: {Macro.underscore(k), v}
    end)
  end
end
Now, when you validate data against this schema, the keys will be transformed to snake_case:
iex> schema = MyApp.User.schema()
iex> Zoi.parse(schema, %{"firstName" => "John", "lastName" => "Doe", "address" => %{"streetAddress" => "21 2nd Street", "city" => "New York"}})
{:ok, %{"first_name" => "John", "last_name" => "Doe", "address" => %{"street_address" => "21 2nd Street", "city" => "New York"}}}
You can also apply key transformation, for example if the data to be validated doesn't really have consistent keys, or if you want to normalize keys before validation.
defmodule MyApp.User do
  @moduledoc false

  def schema() do
    Zoi.object(%{
      "@name" => Zoi.string(),
      "__last_name__" => Zoi.string()
    })
    |> map_to_atom_keys()
  end

  defp map_to_atom_keys(schema) do
    schema
    |> Zoi.transform(fn map ->
      Enum.map(map, fn {k, v} ->
        case k do
          "@name" -> {:name, v}
          "__last_name__" -> {:last_name, v}
          other -> {other, v}
        end
      end)
      |> Enum.into(%{})
    end)
  end
end
Now when you validate data against this schema, the keys will be transformed to the desired format:
iex> schema = MyApp.User.schema()
iex> Zoi.parse(schema, %{"@name" => "John", "__last_name__" => "Doe"})
{:ok, %{name: "John", last_name: "Doe"}}
And the error messages will reflect the parameter keys before transformation:
iex> Zoi.parse(schema, %{"@name" => "John"})
{:error, [%Zoi.Error{message: "is required", path: ["__last_name__"]}]}


  

    Generating Schemas from JSON example

Zoi offers a very flexible way to create schemas and validate data against them. Usually, creating schemas is done programmatically but it can be a tedious task.
In this example, we will demonstrate how to infer schemas from JSON data using Zoi.
Let's say that you need to integrate with an external API and they provide you with a sample JSON response:
{
  "data": {
    "id": "60e59b99c8ca1d58514a2322",
    "project_name": "My Project",
    "description": "Project Description",
    "status": "NEW",
    "meta": {
      "id": 10744,
      "name": "Project 1",
      "slug": "/project-1",
      "symbol": "PRD1"
    },
    "start_date": "2021-06-01T22:11:00.000Z",
    "end_date": "2021-07-01T22:11:00.000Z",
    "total_prize": 20000000000,
    "winner_count": 1000,
    "link": "https://example.com"
  },
  "status": {
    "timestamp": "2025-08-29T06:46:37.240Z",
    "error_code": 0,
    "error_message": "message"
  }
}
We can use Zoi to infer a schema from this JSON data. Here's how you can do it:
defmodule ZoiJsonCodegen do
  @moduledoc """
  Generate Elixir source code (as a string) for a Zoi schema from a JSON example.
  """

  def write_module_from_json(json, module_name) do
    body = code_from_json(json)
    mod_file = Path.join("lib", Macro.underscore("#{module_name}") <> ".ex")

    source = """
    defmodule #{module_name} do
      @moduledoc false

      def #{:schema}() do
        #{body}
      end
    end
    """

    formatted = Code.format_string!(source) |> IO.iodata_to_binary()
    File.write!(mod_file, formatted)
    mod_file
  end

  def code_from_json(json) when is_binary(json) do
    json
    |> Jason.decode!()
    |> build_schema_ast()
    |> ast_to_string()
  end

  # Convert quoted AST to pretty Elixir source string
  defp ast_to_string(ast) do
    ast
    |> Macro.to_string()
    |> Code.format_string!()
    |> IO.iodata_to_binary()
  end

  # Build AST for a Zoi schema call from decoded JSON
  defp build_schema_ast(v) when is_map(v) do
    map_ast =
      {:%{}, [],
       Enum.map(v, fn {k, vv} ->
         {k, build_schema_ast(vv)}
       end)}

    quote(do: Zoi.object(unquote(map_ast)))
  end

  defp build_schema_ast(v) when is_list(v) do
    inner =
      case v do
        [h | _] -> build_schema_ast(h)
        [] -> quote(do: Zoi.any())
      end

    quote(do: Zoi.array(unquote(inner)))
  end

  defp build_schema_ast(v) when is_binary(v), do: quote(do: Zoi.string())
  defp build_schema_ast(v) when is_integer(v), do: quote(do: Zoi.integer())
  defp build_schema_ast(v) when is_float(v), do: quote(do: Zoi.number())
  defp build_schema_ast(v) when is_boolean(v), do: quote(do: Zoi.boolean())
  defp build_schema_ast(_), do: quote(do: Zoi.optional(Zoi.any()))
end
This code might be difficult to understand at first glance, but the key function is build_schema_ast/1, which recursively traverses the structure and builds an Elixir AST representing the corresponding Zoi schema.
Now we can use this module to generate a schema from the provided JSON example:
jason_string = "..." # Replace with the JSON string from the example
ZoiJsonCodegen.write_module_from_json(decoded_json, "MyApp.ExternalApiResponse")
This will result in a new file lib/my_app/external_api_response.ex containing the following schema:
defmodule MyApp.ExternalApiResponse do
  @moduledoc false

  def schema() do
    Zoi.object(%{
      "data" =>
        Zoi.object(%{
          "description" => Zoi.string(),
          "end_date" => Zoi.string(),
          "id" => Zoi.string(),
          "link" => Zoi.string(),
          "meta" =>
            Zoi.object(%{
              "id" => Zoi.integer(),
              "name" => Zoi.string(),
              "slug" => Zoi.string(),
              "symbol" => Zoi.string()
            }),
          "project_name" => Zoi.string(),
          "start_date" => Zoi.string(),
          "status" => Zoi.string(),
          "total_prize" => Zoi.integer(),
          "winner_count" => Zoi.integer()
        }),
      "status" =>
        Zoi.object(%{
          "error_code" => Zoi.integer(),
          "error_message" => Zoi.string(),
          "timestamp" => Zoi.string()
        })
    })
  end
end
This is a fully functional Zoi schema that you can use to validate data received from the external API. The generator is quite basic and may not cover all edge cases, but it provides a solid starting point for generating schemas from JSON examples.

Zoi is a schema validation library for Elixir, designed to provide a simple and flexible way to define and validate data.
It allows you to create schemas for various data types, including strings, integers, booleans, and complex objects, with built-in support for validations like minimum and maximum values, regex patterns, and email formats.
user = Zoi.object(%{
  name: Zoi.string() |> Zoi.min(2) |> Zoi.max(100),
  age: Zoi.integer() |> Zoi.min(18) |> Zoi.max(120),
  email: Zoi.email()
})

Zoi.parse(user, %{
  name: "Alice",
  age: 30,
  email: "alice@email.com"
})
# {:ok, %{name: "Alice", age: 30, email: "alice@email.com"}}
Coercion
By default, Zoi will not attempt to infer input data to match the expected type. For example, if you define a schema that expects a string, passing an integer will result in an error.
iex> Zoi.string() |> Zoi.parse(123)
{:error, [%Zoi.Error{message: "invalid type: must be a string"}]}
If you need coercion, you can enable it by passing the :coerce option:
iex> Zoi.string(coerce: true) |> Zoi.parse(123)
{:ok, "123"}
Custom errors
You can customize parsing error messages the primitive types by passing the error option:
iex> schema = Zoi.integer(error: "must be a number")
iex> Zoi.parse(schema, "a")
{:error, [%Zoi.Error{message: "must be a number"}]}

          @type transform() ::

Defines a schema that accepts any type of input.
This is useful when you want to allow any data type without validation.
Example
iex> schema = Zoi.any()
iex> Zoi.parse(schema, "hello")
{:ok, "hello"}
iex> Zoi.parse(schema, 42)
{:ok, 42}
iex> Zoi.parse(schema, %{key: "value"})
{:ok, %{key: "value"}}

Defines an atom type schema.
Examples
iex> schema = Zoi.atom()
iex> Zoi.parse(schema, :atom)
{:ok, :atom}
iex> Zoi.parse(schema, "not_an_atom")
{:error, [%Zoi.Error{message: "invalid type: must be an atom"}]}

Defines a boolean type schema.
Example
iex> schema = Zoi.boolean()
iex> Zoi.parse(schema, true)
{:ok, true}
For coercion, you can pass the :coerce option:
iex> Zoi.boolean(coerce: true) |> Zoi.parse("true")
{:ok, true}

Defines a float type schema.
Example
iex> schema = Zoi.float()
iex> Zoi.parse(schema, 3.14)
{:ok, 3.14}
Built-in validations for floats include:
Zoi.min(0.0)
Zoi.max(100.0)
For coercion, you can pass the :coerce option:
iex> Zoi.float(coerce: true) |> Zoi.parse("3.14")
{:ok, 3.14}

Defines a number type schema.
Example
iex> shema = Zoi.integer()
iex> Zoi.parse(shema, 42)
{:ok, 42}
iex> Zoi.parse(shema, "42")
{:error, [%Zoi.Error{message: "invalid type: must be an integer"}]}
For coercion, you can pass the :coerce option:
iex> Zoi.integer(coerce: true) |> Zoi.parse("42")
{:ok, 42}

Defines a literal type schema.
This schema only accepts a specific literal value as valid input.
Example
iex> schema = Zoi.literal(true)
iex> Zoi.parse(schema, true)
{:ok, true}
iex> Zoi.parse(schema, :other_value)
{:error, [%Zoi.Error{message: "invalid type: does not match literal"}]}
iex> schema = Zoi.literal(42)
iex> Zoi.parse(schema, 42)
{:ok, 42}
iex> Zoi.parse(schema, 43)
{:error, [%Zoi.Error{message: "invalid type: does not match literal"}]}

Defines a nil type schema.
This schema only accepts nil as valid input.
Example
iex> schema = Zoi.null()
iex> Zoi.parse(schema, nil)
{:ok, nil}
iex> Zoi.parse(schema, "not_nil")
{:error, [%Zoi.Error{message: "invalid type: must be nil"}]}

Defines the numeric type schema.
This type is a union of Zoi.integer() and Zoi.float(), allowing you to validate both integers and floats.
Example
iex> schema = Zoi.number()
iex> Zoi.parse(schema, 42)
{:ok, 42}
iex> Zoi.parse(schema, 3.14)
{:ok, 3.14}
iex> Zoi.parse(schema, "42")
{:error, [%Zoi.Error{message: "invalid type: must be a number"}]}

Defines a string type schema.
Example
iex> schema = Zoi.string()
iex> Zoi.parse(schema, "hello")
{:ok, "hello"}
iex> Zoi.parse(schema, :world)
{:error, [%Zoi.Error{message: "invalid type: must be a string"}]}
Zoi provides built-in validations for strings, such as:
Zoi.gt(100)
Zoi.gte(100)
Zoi.lt(2)
Zoi.lte(2)
Zoi.min(2) # alias for `Zoi.gte(2)`
Zoi.max(100) # alias for `Zoi.lte(100)`
Zoi.starts_with("hello")
Zoi.ends_with("world")
Zoi.length(5)
Zoi.regex(~r/^[a-zA-Z]+$/)
Additionally it can perform data transformation:
Zoi.string()
|> Zoi.trim()
|> Zoi.to_downcase()
|> Zoi.to_uppercase()
for coercion, you can pass the :coerce option:
iex> Zoi.string(coerce: true) |> Zoi.parse(123)
{:ok, "123"}

Defines a string boolean type schema.
This type parses "boolish" string values:
# thruthy values: true, "true", "1", "yes", "on", "y", "enabled"
# falsy values: false, "false", "0", "no", "off", "n", "disabled"
Example
iex> schema = Zoi.string_boolean()
iex> Zoi.parse(schema, "true")
{:ok, true}
iex> Zoi.parse(schema, "false")
{:ok, false}
iex> Zoi.parse(schema, "yes")
{:ok, true}
iex> Zoi.parse(schema, "no")
{:ok, false}
You can also specify custom truthy and falsy values using the :truthy and :falsy options:
iex> schema = Zoi.string_boolean(truthy: ["yes", "y"], falsy: ["no", "n"])
iex> Zoi.parse(schema, "yes")
{:ok, true}
iex> Zoi.parse(schema, "no")
{:ok, false}
By default the string boolean type is case insensitive and the input is converted to lowercase during the comparison. You can change this behavior using the :case option:
iex> schema = Zoi.string_boolean(case: "sensitive")
iex> Zoi.parse(schema, "True")
{:error, [%Zoi.Error{message: "invalid type: must be a string boolean"}]}
iex> Zoi.parse(schema, "true")
{:ok, true}

Defines a array type schema.
Use Zoi.array(elements) to define an array of a specific type:
iex> schema = Zoi.array(Zoi.string())
iex> Zoi.parse(schema, ["hello", "world"])
{:ok, ["hello", "world"]}
iex> Zoi.parse(schema, ["hello", 123])
{:error, [%Zoi.Error{message: "invalid type: must be a string", path: [1]}]}
Built-in validations for integers include:
Zoi.gt(5)
Zoi.gte(5)
Zoi.lt(2)
Zoi.lte(2)
Zoi.min(2) # alias for `Zoi.gte/1`
Zoi.max(5) # alias for `Zoi.lte/1`
Zoi.length(5)

Defines an enum type schema.
Use Zoi.enum(values) to define a schema that accepts only specific values:
iex> schema = Zoi.enum([:red, :green, :blue])
iex> Zoi.parse(schema, :red)
{:ok, :red}
iex> Zoi.parse(schema, :yellow)
{:error, [%Zoi.Error{message: "invalid option, must be one of: red, green, blue"}]}
You can also specify enum as strings:
iex> schema = Zoi.enum(["red", "green", "blue"])
iex> Zoi.parse(schema, "red")
{:ok, "red"}
iex> Zoi.parse(schema, "yellow")
{:error, [%Zoi.Error{message: "invalid option, must be one of: red, green, blue"}]}
or with key-value pairs:
iex> schema = Zoi.enum([red: "Red", green: "Green", blue: "Blue"])
iex> Zoi.parse(schema, "Red")
{:ok, :red}
iex> Zoi.parse(schema, "Yellow")
{:error, [%Zoi.Error{message: "invalid option, must be one of: Red, Green, Blue"}]}
Integer values can also be used:
iex> schema = Zoi.enum([1, 2, 3])
iex> Zoi.parse(schema, 1)
{:ok, 1}
iex> Zoi.parse(schema, 4)
{:error, [%Zoi.Error{message: "invalid option, must be one of: 1, 2, 3"}]}
And Integers with key-value pairs also is allowed:
iex> schema = Zoi.enum([one: 1, two: 2, three: 3])
iex> Zoi.parse(schema, 1)
{:ok, :one}
iex> Zoi.parse(schema, 4)
{:error, [%Zoi.Error{message: "invalid value for enum"}]}
{:error, [%Zoi.Error{message: "invalid option, must be one of: 1, 2, 3"}]}

Extends two object type schemas into one.
This function merges the fields of two object schemas. If there are overlapping fields, the fields from the second schema will override those from the first.
Example
iex> user = Zoi.object(%{name: Zoi.string()})
iex> role = Zoi.object(%{role: Zoi.enum([:admin,:user])})
iex> user_with_role = Zoi.extend(user, role)
iex> Zoi.parse(user_with_role, %{name: "Alice", role: :admin})
{:ok, %{name: "Alice", role: :admin}}

Defines a keyword list type schema.
iex> schema = Zoi.keyword(name: Zoi.string(), age: Zoi.integer())
iex> Zoi.parse(schema, [name: "Alice", age: 30])
{:ok, [name: "Alice", age: 30]}
iex> Zoi.parse(schema, %{name: "Alice", age: 30})
{:error, [%Zoi.Error{message: "invalid type: must be a keyword list"}]}
By default, unrecognized keys will be removed from the parsed data. If you want to not allow unrecognized keys, use the :strict option:
iex> schema = Zoi.keyword([name: Zoi.string()], strict: true)
iex> Zoi.parse(schema, [name: "Alice", age: 30])
{:error, [%Zoi.Error{message: "unrecognized key: 'age'"}]}
All fields are optional by default in keyword lists, but you can make them required by using Zoi.required/1:
iex> schema = Zoi.keyword([name: Zoi.string() |> Zoi.required()])
iex> Zoi.parse(schema, [])
{:error, [%Zoi.Error{message: "is required", path: [:name]}]}

alias for Zoi.array/2

Defines a map type schema with Zoi.any() type.
This type is the same as creating the following map:
Zoi.map(Zoi.any(), Zoi.any())

Defines a map type schema.
Example
iex> schema = Zoi.map(Zoi.string(), Zoi.integer())
iex> Zoi.parse(schema, %{"a" => 1, "b" => 2})
{:ok, %{"a" => 1, "b" => 2}}
iex> Zoi.parse(schema, %{"a" => "1", "b" => 2})
{:error, [%Zoi.Error{message: "invalid type: must be an integer", path: ["a"]}]}

Defines a object type schema.
Use Zoi.object(fields) to define complex objects with nested schemas:
iex> user_schema = Zoi.object(%{
...> name: Zoi.string() |> Zoi.min(2) |> Zoi.max(100),
...> age: Zoi.integer() |> Zoi.min(18) |> Zoi.max(120),
...> email: Zoi.email()
...> })
iex> Zoi.parse(user_schema, %{name: "Alice", age: 30, email: "alice@email.com"})
{:ok, %{name: "Alice", age: 30, email: "alice@email.com"}}
By default all fields are required, but you can make them optional by using Zoi.optional/1:
iex> user_schema = Zoi.object(%{
...> name: Zoi.string() |> Zoi.optional(),
...> age: Zoi.integer() |> Zoi.optional(),
...> email: Zoi.email() |> Zoi.optional()
...> })
iex> Zoi.parse(user_schema, %{name: "Alice"})
{:ok, %{name: "Alice"}}
By default, unrecognized keys will be removed from the parsed data. If you want to not allow unrecognized keys, use the :strict option:
iex> schema = Zoi.object(%{name: Zoi.string()}, strict: true)
iex> Zoi.parse(schema, %{name: "Alice", age: 30})
{:error, [%Zoi.Error{message: "unrecognized key: 'age'"}]}
String keys and Atom keys
Objects can be declared using string keys too, this would set the expectation that the param data is also using string keys:
iex> schema = Zoi.object(%{"name" => Zoi.string()})
iex> Zoi.parse(schema, %{"name" => "Alice"})
{:ok, %{"name" => "Alice"}}
iex> Zoi.parse(schema, %{name: "Alice"})
{:error, [%Zoi.Error{message: "is required", path: ["name"]}]}
It's possible coerce the keys to atoms using the :coerce option:
iex> schema = Zoi.object(%{name: Zoi.string()}, coerce: true)
iex> Zoi.parse(schema, %{"name" => "Alice"})
{:ok, %{name: "Alice"}}
Which will automatically convert string keys to atom keys.
Nullable vs Optional fields
The Zoi.optional/1 function makes a field optional, meaning it can be omitted from the input data. If the field is not present, it will not be included in the parsed result.
The Zoi.nullable/1 function allows a field to be nil, meaning it can be explicitly set to nil in the input data. If the field is set to nil, it will be included in the parsed result as nil.
iex> schema = Zoi.object(%{name: Zoi.string() |> Zoi.optional(), age: Zoi.integer() |> Zoi.nullable()})
iex> Zoi.parse(schema, %{name: "Alice", age: nil})
{:ok, %{name: "Alice", age: nil}}
iex> Zoi.parse(schema, %{name: "Alice"})
{:error, [%Zoi.Error{message: "is required", path: [:age]}]}
Optional vs Default fields
There are two options to define the behaviour of a field being optional and with a default value:
	If the field is not present in the input data OR nil, it will be included in the parsed result with the default value.
	If the field not present in the input data, it will not be included on the parsed result. If the value is nil, it will be included in the parsed result with the default value.

The order you encapsulate the type matters, to implement the first option, the encapsulation should be Zoi.default(Zoi.optional(type, default_value)):
iex> schema = Zoi.object(%{name: Zoi.default(Zoi.optional(Zoi.string()), "default value")})
iex> Zoi.parse(schema, %{})
{:ok, %{name: "default value"}}
iex> Zoi.parse(schema, %{name: nil})
{:ok, %{name: "default value"}}
The second option is implemented by encapsulating the type as Zoi.optional(Zoi.default(type, default_value)):
iex> schema = Zoi.object(%{name: Zoi.optional(Zoi.default(Zoi.string(), "default value"))})
iex> Zoi.parse(schema, %{})
{:ok, %{}}
iex> Zoi.parse(schema, %{name: nil})
{:ok, %{name: "default value"}}

Defines a struct type schema.
This type is similar to Zoi.object/2, but it is specifically designed to work with Elixir structs.
Example
defmodule MyApp.User do
  defstruct [:name, :age, :email]
end

schema = Zoi.struct(MyApp.User, %{
  name: Zoi.string() |> Zoi.min(2) |> Zoi.max(100),
  age: Zoi.integer() |> Zoi.min(18) |> Zoi.max(120),
  email: Zoi.email()
})
Zoi.parse(schema, %MyApp.User{name: "Alice", age: 30, email: "alice@email.com"})
#=> {:ok, %MyApp.User{name: "Alice", age: 30, email: "alice@email.com"}}
Zoi.parse(schema, %{})
#=> {:error, "invalid type: must be a struct"}
By default, all fields are required, but you can make them optional by using Zoi.optional/1:
schema = Zoi.struct(MyApp.User, %{
  name: Zoi.string() |> Zoi.optional(),
  age: Zoi.integer() |> Zoi.optional(),
  email: Zoi.email() |> Zoi.optional()
})
Zoi.parse(schema, %MyApp.User{name: "Alice"})
#=> {:ok, %MyApp.User{name: "Alice"}}
You can also parse maps into structs by enabling the :coerce option:
schema = Zoi.struct(MyApp.User, %{
  name: Zoi.string(),
  age: Zoi.integer(),
  email: Zoi.email()
}, coerce: true)
Zoi.parse(schema, %{name: "Alice", age: 30, email: "alice@email.com"})
#=> {:ok, %MyApp.User{name: "Alice", age: 30, email: "alice@email.com"}}
# Also with string keys
Zoi.parse(schema, %{"name" => "Alice", "age" => 30, "email" => "alice@email.com"})
#=> {:ok, %MyApp.User{name: "Alice", age: 30, email: "alice@email.com"}}

Defines a tuple type schema.
Use Zoi.tuple(fields) to define a tuple with specific types for each element:
iex> schema = Zoi.tuple({Zoi.string(), Zoi.integer()})
iex> Zoi.parse(schema, {"hello", 42})
{:ok, {"hello", 42}}
iex> Zoi.parse(schema, {"hello", "world"})
{:error, [%Zoi.Error{message: "invalid type: must be an integer", path: [1]}]}

      default(inner, value, opts \\ [])



        
          
        

    

  


  

Creates a default value for the schema.
This allows you to specify a default value that will be used if the input is nil or not provided.
Example
iex> schema = Zoi.string() |> Zoi.default("default value")
iex> Zoi.parse(schema, nil)
{:ok, "default value"}

Defines an intersection type schema.
An intersection type allows you to combine multiple schemas into one, requiring that the input data satisfies all of them.
Example
iex> schema = Zoi.intersection([
...>   Zoi.string() |> Zoi.min(2),
...>   Zoi.string() |> Zoi.max(5)
...> ])
iex> Zoi.parse(schema, "helloworld")
{:error, [%Zoi.Error{message: "too big: must have at most 5 characters"}]}
iex> Zoi.parse(schema, "hi")
{:ok, "hi"}
If you define the validation on the intersection itself, it will apply to all types in the intersection:
iex> schema = Zoi.intersection([
...>   Zoi.string(),
...>   Zoi.integer(coerce: true)
...> ]) |> Zoi.min(3)
iex> Zoi.parse(schema, "115")
{:ok, 115}
iex> Zoi.parse(schema, "2")
{:error, [%Zoi.Error{message: "too small: must have at least 3 characters"}]}

Defines a schema that allows nil values.
Examples
iex> schema = Zoi.string() |> Zoi.nullable()
iex> Zoi.parse(schema, nil)
{:ok, nil}
iex> Zoi.parse(schema, "hello")
{:ok, "hello"}

Makes the schema optional for the Zoi.object/2 and Zoi.keyword/2 types.
Example
iex> schema = Zoi.object(%{name: Zoi.string() |> Zoi.optional()})
iex> Zoi.parse(schema, %{})
{:ok, %{}}

Makes the schema required for the Zoi.object/2 and Zoi.keyword/2 types.
Example
iex> schema = Zoi.keyword([name: Zoi.string() |> Zoi.required()])
iex> Zoi.parse(schema, [])
{:error, [%Zoi.Error{message: "is required", path: [:name]}]}

Defines a union type schema.
Example
iex> schema = Zoi.union([Zoi.string(), Zoi.integer()])
iex> Zoi.parse(schema, "hello")
{:ok, "hello"}
iex> Zoi.parse(schema, 42)
{:ok, 42}
iex> Zoi.parse(schema, true)
{:error, [%Zoi.Error{message: "invalid type: must be an integer"}]}
This type also allows to define validations for each type in the union:
iex> schema = Zoi.union([
...>   Zoi.string() |> Zoi.min(2),
...>   Zoi.integer() |> Zoi.min(0)
...> ])
iex> Zoi.parse(schema, "h") # fails on string and try to parse as integer
{:error, [%Zoi.Error{message: "invalid type: must be an integer"}]}
iex> Zoi.parse(schema, -1)
{:error, [%Zoi.Error{message: "too small: must be at least 0"}]}
If you define the validation on the union itself, it will apply to all types in the union:
iex> schema = Zoi.union([
...>   Zoi.string(),
...>   Zoi.integer()
...> ]) |> Zoi.min(3)
iex> Zoi.parse(schema, "hello")
{:ok, "hello"}
iex> Zoi.parse(schema, 2)
{:error, [%Zoi.Error{message: "too small: must be at least 3"}]}

  


        

      

      
        Extensions


        


  
    
      
    
    
      refine(schema, fun)



        
          
        

    

  


  

      

          @spec refine(schema :: Zoi.Type.t(), fun :: refinement()) :: Zoi.Type.t()


      


Adds a custom validation function to the schema.
This function will be called with the input data and options, and should return :ok for valid data or {:error, reason} for invalid data.
iex> schema = Zoi.string() |> Zoi.refine(fn value -> 
...>   if String.length(value) > 5, do: :ok, else: {:error, "must be longer than 5 characters"}
...> end)
iex> Zoi.parse(schema, "hello")
{:error, [%Zoi.Error{message: "must be longer than 5 characters"}]}
iex> Zoi.parse(schema, "hello world")
{:ok, "hello world"}
Returning multiple errors
You can use the context when defining the Zoi.refine/2 function to return multiple errors.
iex> schema = Zoi.string() |> Zoi.refine(fn value, ctx ->
...>   if String.length(value) < 5 do
...>     Zoi.Context.add_error(ctx, "must be longer than 5 characters")
...>     |> Zoi.Context.add_error("must be shorter than 10 characters")
...>   end
...> end)
iex> Zoi.parse(schema, "hi")
{:error, [
  %Zoi.Error{message: "must be longer than 5 characters"},
  %Zoi.Error{message: "must be shorter than 10 characters"}
]}
mfa
You can also pass a mfa (module, function, args) to the Zoi.refine/2 function. This is recommended if
you are declaring schemas during compile time:
defmodule MySchema do
  use Zoi

  @schema Zoi.string() |> Zoi.refine({__MODULE__, :validate, []})

  def validate(value, opts \\ []) do
    if String.length(value) > 5 do
      :ok
    else 
      {:error, "must be longer than 5 characters"}
    end
  end
end
Since during the module compilation, anonymous functions are not available, you can use the mfa option to pass a module, function and arguments.
The opts argument is mandatory, this is where the ctx is passed to the function and you can leverage the Zoi.Context to add extra errors.
In general, most cases the :ok or {:error, reason} returns will be enough. Use the context only if you need extra errors or modify the context in some way.

          @spec transform(schema :: Zoi.Type.t(), fun :: transform()) :: Zoi.Type.t()


      


Adds a transformation function to the schema.
This function will be applied to the input data after parsing but before validations.
Example
iex> schema = Zoi.string() |> Zoi.transform(fn value ->
...>   {:ok, String.trim(value)}
...> end)
iex> Zoi.parse(schema, "  hello world  ")
{:ok, "hello world"}
You can also use mfa (module, function, args) to pass a transformation function:
iex> defmodule MyTransforms do
...>   def trim(value, _opts) do
...>     {:ok, String.trim(value)}
...>   end
...> end
iex> schema = Zoi.string() |> Zoi.transform({MyTransforms, :trim, []})
iex> Zoi.parse(schema, "  hello world  ")
{:ok, "hello world"}
This is useful if you are defining schemas at compile time, where anonymous functions are not available.
The opts argument is mandatory, this is where the ctx is passed to the function and you can leverage the Zoi.Context to add extra errors.
In general, most cases the {:ok, value} or {:error, reason} returns will be enough. Use the context only if you need extra errors or modify the context in
some way.
Using context for validation
You can use the context when defining the Zoi.transform/2 function to return multiple errors.
iex> schema = Zoi.string() |> Zoi.transform(fn value, ctx ->
...>   if String.length(value) < 5 do
...>     Zoi.Context.add_error(ctx, "must be longer than 5 characters")
...>     |> Zoi.Context.add_error("must be shorter than 10 characters")
...>   else
...>     {:ok, String.trim(value)}
...>   end
...> end)
iex> Zoi.parse(schema, "hi")
{:error, [
  %Zoi.Error{message: "must be longer than 5 characters"},
  %Zoi.Error{message: "must be shorter than 10 characters"}
]}
The ctx is a Zoi.Context struct that contains information about the current parsing context, including the path, options, and any errors that have been added so far.

          @spec email() :: Zoi.Type.t()


      


Validates that the string is a valid email format.
Example
iex> schema = Zoi.email()
iex> Zoi.parse(schema, "test@test.com")
{:ok, "test@test.com"}
iex> Zoi.parse(schema, "invalid-email")
{:error, [%Zoi.Error{message: "invalid email format"}]}
It uses a regex pattern to validate the email format, which checks for a standard email structure including local part, domain, and top-level domain:
~r/^(?!.)(?!.*..)([a-z0-9_'+-.]*)[a-z0-9_+-]@([a-z0-9][a-z0-9-]*.)+[a-z]{2,}$/i

Validates that the string is a valid hexadecimal format.
Example
iex> schema = Zoi.hex()
iex> Zoi.parse(schema, "a3c113")
{:ok, "a3c113"}
iex> Zoi.parse(schema, "invalid-hex")
{:error, [%Zoi.Error{message: "invalid hex format"}]}

Validates that the string is a valid IPv4 address.
Example
iex> schema = Zoi.ipv4()
iex> Zoi.parse(schema, "127.0.0.1")
{:ok, "127.0.0.1"}

Validates that the string is a valid IPv6 address.
Example
iex> schema = Zoi.ipv6()
iex> Zoi.parse(schema, "2001:0db8:85a3:0000:0000:8a2e:0370:7334")
{:ok, "2001:0db8:85a3:0000:0000:8a2e:0370:7334"}
iex> Zoi.parse(schema, "invalid-ipv6")
{:error, [%Zoi.Error{message: "invalid IPv6 address"}]}

Defines a URL format validation.
Example
iex> schema = Zoi.url()
iex> Zoi.parse(schema, "https://example.com")
{:ok, "https://example.com"}
iex> Zoi.parse(schema, "invalid-url")
{:error, [%Zoi.Error{message: "invalid URL"}]}

          @spec uuid(opts :: keyword()) :: Zoi.Type.t()


      


Validates that the string is a valid UUID format.
You can specify the UUID version using the :version option, which can be one of "v1", "v2", "v3", "v4", "v5", "v6", "v7", or "v8". If no version is specified, it defaults to any valid UUID format.
Example
iex> schema = Zoi.uuid()
iex> Zoi.parse(schema, "550e8400-e29b-41d4-a716-446655440000")
{:ok, "550e8400-e29b-41d4-a716-446655440000"}
iex> Zoi.parse(schema, "invalid-uuid")
{:error, [%Zoi.Error{message: "invalid UUID format"}]}

iex> schema = Zoi.uuid(version: "v8")
iex> Zoi.parse(schema, "6d084cef-a067-8e9e-be6d-7c5aefdfd9b4")
{:ok, "6d084cef-a067-8e9e-be6d-7c5aefdfd9b4"}

      This function is deprecated. Use Zoi.metadata/1 instead.
    


  

      

          @spec example(schema :: Zoi.Type.t()) :: input()


      


Retrieves an example value from the schema. If no example is defined, it returns nil.
Example
iex> schema = Zoi.string(example: "example string")
iex> Zoi.example(schema)
"example string"
This directive is specally useful for documentation and testing purposes.
As an example, you can define a schema as it follows:
defmodule MyApp.UserSchema do
  @schema Zoi.object(
            %{
              name: Zoi.string() |> Zoi.min(2) |> Zoi.max(100),
              age: Zoi.integer() |> Zoi.optional()
            },
            example: %{name: "Alice", age: 30}
          )

  def schema, do: @schema
end
Then you can test if the example matches the schema:
defmodule MyApp.UserSchemaTest do
  use ExUnit.Case
  alias MyApp.UserSchema

  test "example matches schema" do
    example = Zoi.example(UserSchema.schema())
    assert {:ok, example} == Zoi.parse(UserSchema.schema(), example)
  end
end

          @spec metadata(schema :: Zoi.Type.t()) :: keyword()


      


Retrieves the metadata associated with the schema.
It's often useful to store additional information about the schema, such as descriptions, titles, or custom identifiers.
Example
iex> schema = Zoi.string(metadata: [id: "1", description: "A simple string"])
iex> Zoi.metadata(schema)
[id: "1", description: "A simple string"]
You can also add an example helper that can be used on your tests and documentation:
defmodule MyApp.UserSchema do
  @schema Zoi.object(
            %{
              name: Zoi.string() |> Zoi.min(2) |> Zoi.max(100),
              age: Zoi.integer() |> Zoi.optional()
            },
            metadata: [
              example: %{name: "Alice", age: 30},
              doc: "A user schema with name and optional age",
              moduledoc: "Schema representing a user with name and optional age"
            ]
          )
  @moduledoc """
  #{Zoi.metadata(@schema)[:moduledoc]}
  """

  @doc """
  #{Zoi.metadata(@schema)[:doc]}
  """
  def schema, do: @schema
end

defmodule MyApp.UserSchemaTest do
  use ExUnit.Case
  alias MyApp.UserSchema

  test "example matches schema" do
    example = Zoi.metadata(UserSchema.schema())[:example]
    assert {:ok, example} == Zoi.parse(UserSchema.schema(), example)
  end
end

          @spec parse(schema :: Zoi.Type.t(), input :: input(), opts :: options()) :: result()


      


Parse input data against a schema.
Accepts optional coerce: true option to enable coercion.
Examples
iex> schema = Zoi.string() |> Zoi.min(2) |> Zoi.max(100)
iex> Zoi.parse(schema, "hello")
{:ok, "hello"}
iex> Zoi.parse(schema, "h")
{:error, [%Zoi.Error{message: "too small: must have at least 2 characters"}]}
iex> Zoi.parse(schema, 123, coerce: true)
{:ok, "123"}

          @spec parse!(schema :: Zoi.Type.t(), input :: input(), opts :: options()) :: any()


      


Similar to Zoi.parse/3, but raises an error if parsing fails.
Examples
schema = Zoi.string() |> Zoi.min(2) |> Zoi.max(100)
Zoi.parse!(schema, "hello")
#=> "hello"
Zoi.parse!(schema, "h")
# ** (Zoi.ParseError) Parsing error:
#
# too small: must have at least 2 characters

  



  
    
      
    
    
      prettify_errors(errors)



        
          
        

    

  


  

      

          @spec prettify_errors([Zoi.Error.t() | binary()]) :: binary()


      


Converts a list of errors into a human-readable string format.
Each error is displayed on a new line, with its message and path.
Example
iex> errors = [
...>   %Zoi.Error{path: ["name"], message: "is required"},
...>   %Zoi.Error{path: ["age"], message: "invalid type: must be an integer"},
...>   %Zoi.Error{path: ["address", "city"], message: "is required"}
...> ]
iex> Zoi.prettify_errors(errors)
"is required, at name\ninvalid type: must be an integer, at age\nis required, at address.city"

iex> errors = [%Zoi.Error{message: "invalid type: must be a string"}]
iex> Zoi.prettify_errors(errors)
"invalid type: must be a string"

  



  
    
      
    
    
      to_json_schema(schema)



        
          
        

    

  


  

See Zoi.JSONSchema

          @spec treefy_errors([Zoi.Error.t()]) :: map()


      


Converts a list of errors into a tree structure, where each error is placed at its corresponding path.
This is useful for displaying validation errors in a structured way, such as in a form.
Example
iex> errors = [
...>   %Zoi.Error{path: ["name"], message: "is required"},
...>   %Zoi.Error{path: ["age"], message: "invalid type: must be an integer"},
...>   %Zoi.Error{path: ["address", "city"], message: "is required"}
...> ]
iex> Zoi.treefy_errors(errors)
%{
  "name" => ["is required"],
  "age" => ["invalid type: must be an integer"],
  "address" => %{
    "city" => ["is required"]
  }
}
If you use this function on types without path (like Zoi.string()), it will create a top-level :__errors__ key:
iex> errors = [%Zoi.Error{message: "invalid type: must be a string"}]
iex> Zoi.treefy_errors(errors)
%{__errors__: ["invalid type: must be a string"]}
Errors without a path are considered top-level errors and are grouped under :__errors__.
This is how Zoi also handles errors when Zoi.object/2 is used with :strict option, where unrecognized keys are added to the :__errors__ key.

          @spec type_spec(schema :: Zoi.Type.t(), opts :: options()) :: Macro.t()


      


Generates the Elixir type specification for a given schema.
Example
defmodule MyApp.Schema do
  @schema Zoi.string() |> Zoi.min(2) |> Zoi.max(100)
  @type t :: unquote(Zoi.type_spec(@schema))
end
This will generate the following type specification:
@type t :: binary()
This also applies to complex types, such as Zoi.object/2:
defmodule MyApp.User do
  @schema Zoi.object(%{
    name: Zoi.string() |> Zoi.min(2) |> Zoi.max(100),
    age: Zoi.integer() |> Zoi.optional(),
    email: Zoi.email()
  })
  @type t :: unquote(Zoi.type_spec(@schema))
end
Which will generate:
@type t :: %{
  required(:name) => binary(),
  optional(:age) => integer(),
  required(:email) => binary()
}
Union types are also supported:
  Zoi.union([Zoi.string(), Zoi.integer()])
  #=> binary() | integer()
All the types provided by Zoi supports the type spec generation.

          @spec downcase(schema :: Zoi.Type.t()) :: Zoi.Type.t()


      


Validates that a string is in downcase.
Example
iex> schema = Zoi.string() |> Zoi.downcase()
iex> Zoi.parse(schema, "hello world")
{:ok, "hello world"}
iex> Zoi.parse(schema, "Hello World")
{:error, [%Zoi.Error{message: "invalid string: must be lowercase"}]}

          @spec ends_with(schema :: Zoi.Type.t(), suffix :: binary(), opts :: options()) ::
  Zoi.Type.t()


      


Validates that a string ends with a specific suffix.
Example
iex> schema = Zoi.string() |> Zoi.ends_with("world")
iex> Zoi.parse(schema, "hello world")
{:ok, "hello world"}
iex> Zoi.parse(schema, "hello")
{:error, [%Zoi.Error{message: "invalid string: must end with 'world'"}]}

  



    

  
    
      
    
    
      gt(schema, gt, opts \\ [])



        
          
        

    

  


  

      

          @spec gt(schema :: Zoi.Type.t(), gt :: non_neg_integer(), opts :: options()) ::
  Zoi.Type.t()


      


Validates that the input is greater than a specific value.
This can be used for strings, integers, floats and numbers.
Example
iex> schema = Zoi.integer() |> Zoi.gt(2)
iex> Zoi.parse(schema, 3)
{:ok, 3}
iex> Zoi.parse(schema, 2)
{:error, [%Zoi.Error{message: "too small: must be greater than 2"}]}

  



    

  
    
      
    
    
      gte(schema, gte, opts \\ [])



        
          
        

    

  


  

      

          @spec gte(schema :: Zoi.Type.t(), min :: non_neg_integer(), opts :: options()) ::
  Zoi.Type.t()


      


Validates that the input is greater than or equal to a value.
This can be used for strings, integers, floats and numbers.
Example
iex> schema = Zoi.string() |> Zoi.gte(3)
iex> Zoi.parse(schema, "hello")
{:ok, "hello"}
iex> Zoi.parse(schema, "hi")
{:error, [%Zoi.Error{message: "too small: must have at least 3 characters"}]}

          @spec length(schema :: Zoi.Type.t(), length :: non_neg_integer(), opts :: options()) ::
  Zoi.Type.t()


      


Validates that the string has a specific length.
Example
iex> schema = Zoi.string() |> Zoi.length(5)
iex> Zoi.parse(schema, "hello")
{:ok, "hello"}
iex> Zoi.parse(schema, "hi")
{:error, [%Zoi.Error{message: "invalid length: must have 5 characters"}]}

  



    

  
    
      
    
    
      lt(schema, lt, opts \\ [])



        
          
        

    

  


  

      

          @spec lt(schema :: Zoi.Type.t(), lt :: non_neg_integer(), opts :: options()) ::
  Zoi.Type.t()


      


Validates that the input is less than a specific value.
This can be used for strings, integers, floats and numbers.
Example
iex> schema = Zoi.integer() |> Zoi.lt(10)
iex> Zoi.parse(schema, 5)
{:ok, 5}
iex> Zoi.parse(schema, 10)
{:error, [%Zoi.Error{message: "too big: must be less than 10"}]}

  



    

  
    
      
    
    
      lte(schema, lte, opts \\ [])



        
          
        

    

  


  

      

          @spec lte(schema :: Zoi.Type.t(), lte :: non_neg_integer(), opts :: options()) ::
  Zoi.Type.t()


      


Validates that the input is less than or equal to a value.
This can be used for strings, integers, floats and numbers.
Example
iex> schema = Zoi.string() |> Zoi.lte(5)
iex> Zoi.parse(schema, "hello")
{:ok, "hello"}
iex> Zoi.parse(schema, "hello world")
{:error, [%Zoi.Error{message: "too big: must have at most 5 characters"}]}

  



    

  
    
      
    
    
      max(schema, max, opts \\ [])



        
          
        

    

  


  

      

          @spec max(schema :: Zoi.Type.t(), max :: non_neg_integer(), opts :: options()) ::
  Zoi.Type.t()


      


alias for Zoi.lte/2

  



    

  
    
      
    
    
      min(schema, min, opts \\ [])



        
          
        

    

  


  

      

          @spec min(schema :: Zoi.Type.t(), min :: non_neg_integer(), opts :: options()) ::
  Zoi.Type.t()


      


alias for Zoi.gte/2

  



    

  
    
      
    
    
      negative(schema, opts \\ [])



        
          
        

    

  


  

      

          @spec negative(schema :: Zoi.Type.t(), opts :: options()) :: Zoi.Type.t()


      


Validates that the input is a negative number (less than 0).
This can be used for integers, floats and numbers.
Example
iex> schema = Zoi.integer() |> Zoi.negative()
iex> Zoi.parse(schema, -5)
{:ok, -5}
iex> Zoi.parse(schema, 0)
{:error, [%Zoi.Error{message: "too big: must be less than 0"}]}

  



    

  
    
      
    
    
      positive(schema, opts \\ [])



        
          
        

    

  


  

      

          @spec positive(schema :: Zoi.Type.t(), opts :: options()) :: Zoi.Type.t()


      


Validates that the input is a positive number (greater than 0).
This can be used for integers, floats and numbers.
Example
iex> schema = Zoi.integer() |> Zoi.positive()
iex> Zoi.parse(schema, 5)
{:ok, 5}
iex> Zoi.parse(schema, 0)
{:error, [%Zoi.Error{message: "too small: must be greater than 0"}]}

  



    

  
    
      
    
    
      regex(schema, regex, opts \\ [])



        
          
        

    

  


  

      

          @spec regex(schema :: Zoi.Type.t(), regex :: Regex.t(), opts :: options()) ::
  Zoi.Type.t()


      


Validates that the input matches a given regex pattern.
Example
iex> schema = Zoi.string() |> Zoi.regex(~r/^\d+$/)
iex> Zoi.parse(schema, "12345")
{:ok, "12345"}

  



    

  
    
      
    
    
      starts_with(schema, prefix, opts \\ [])



        
          
        

    

  


  

      

          @spec starts_with(schema :: Zoi.Type.t(), prefix :: binary(), opts :: options()) ::
  Zoi.Type.t()


      


Validates that a string starts with a specific prefix.
Example
iex> schema = Zoi.string() |> Zoi.starts_with("hello")
iex> Zoi.parse(schema, "hello world")
{:ok, "hello world"}
iex> Zoi.parse(schema, "world hello")
{:error, [%Zoi.Error{message: "invalid string: must start with 'hello'"}]}

  



  
    
      
    
    
      upcase(schema)



        
          
        

    

  


  

      

          @spec upcase(schema :: Zoi.Type.t()) :: Zoi.Type.t()


      


Validates that a string is in upcase.
Example
iex> schema = Zoi.string() |> Zoi.upcase()
iex> Zoi.parse(schema, "HELLO")
{:ok, "HELLO"}
iex> Zoi.parse(schema, "Hello")
{:error, [%Zoi.Error{message: "invalid string: must be uppercase"}]}

      date(opts \\ [])



        
          
        

    

  


  

Defines a date type schema.
This type is used to validate and parse date values. It will convert the input to a Date structure.
Example
iex> schema = Zoi.date()
iex> Zoi.parse(schema, ~D[2000-01-01])
{:ok, ~D[2000-01-01]}
iex> Zoi.parse(schema, "2000-01-01")
{:error, [%Zoi.Error{message: "invalid type: must be a date"}]}
You can also specify the :coerce option to allow coercion from strings or integers:
iex> schema = Zoi.date(coerce: true)
iex> Zoi.parse(schema, "2000-01-01")
{:ok, ~D[2000-01-01]}
iex> Zoi.parse(schema, 730_485) # 730_485 is the number of days since epoch
{:ok, ~D[2000-01-01]}

  



    

  
    
      
    
    
      datetime(opts \\ [])



        
          
        

    

  


  

Defines a DateTime type schema.
This type is used to validate and parse DateTime values. It will convert the input to a DateTime structure.
Example
iex> schema = Zoi.datetime()
iex> Zoi.parse(schema, ~U[2000-01-01 12:34:56Z])
{:ok, ~U[2000-01-01 12:34:56Z]}
iex> Zoi.parse(schema, "2000-01-01T12:34:56Z")
{:error, [%Zoi.Error{message: "invalid type: must be a datetime"}]}
You can also specify the :coerce option to allow coercion from strings or integers:
iex> schema = Zoi.datetime(coerce: true)
iex> Zoi.parse(schema, "2000-01-01T12:34:56Z")
{:ok, ~U[2000-01-01 12:34:56Z]}
iex> Zoi.parse(schema, 1_464_096_368) # 1_464_096_368 is the Unix timestamp
{:ok, ~U[2016-05-24 13:26:08Z]}

  



    

  
    
      
    
    
      decimal(opts \\ [])



        
          
        

    

  


  

Defines a decimal type schema.
This type is used to validate and parse decimal numbers, which can be useful for financial calculations or precise numeric values.
It uses the Decimal library for handling decimal numbers. It will convert the input to a Decimal structure.
Example
iex> schema = Zoi.decimal()
iex> Zoi.parse(schema, Decimal.new("123.45"))
{:ok, Decimal.new("123.45")}
iex> Zoi.parse(schema, "invalid-decimal")
{:error, [%Zoi.Error{message: "invalid type: must be a decimal"}]}
You can also specify the :coerce option to allow coercion from strings or integers:
iex> schema = Zoi.decimal(coerce: true)
iex> Zoi.parse(schema, "123.45")
{:ok, Decimal.new("123.45")}
iex> Zoi.parse(schema, 123)
{:ok, Decimal.new("123")}

  



    

  
    
      
    
    
      naive_datetime(opts \\ [])



        
          
        

    

  


  

Defines a NaiveDateTime type schema.
This type is used to validate and parse NaiveDateTime values. It will convert the input to a NaiveDateTime structure.
Example
iex> schema = Zoi.naive_datetime()
iex> Zoi.parse(schema, ~N[2000-01-01 23:00:07])
{:ok, ~N[2000-01-01 23:00:07]}
iex> Zoi.parse(schema, "2000-01-01T12:34:56")
{:error, [%Zoi.Error{message: "invalid type: must be a naive datetime"}]}
You can also specify the :coerce option to allow coercion from strings or integers:
iex> schema = Zoi.naive_datetime(coerce: true)
iex> Zoi.parse(schema, "2000-01-01T12:34:56")
{:ok, ~N[2000-01-01 12:34:56]}
iex> Zoi.parse(schema, 1) # 1  is the number of days since epoch
{:ok, ~N[0000-01-01 00:00:01]}

  



    

  
    
      
    
    
      time(opts \\ [])



        
          
        

    

  


  

Defines a time type schema.
This type is used to validate and parse time values. It will convert the input to a Time structure.
Example
iex> schema = Zoi.time()
iex> Zoi.parse(schema, ~T[12:34:56])
{:ok, ~T[12:34:56]}
iex> Zoi.parse(schema, "12:34:56")
{:error, [%Zoi.Error{message: "invalid type: must be a time"}]}
You can also specify the :coerce option to allow coercion from strings:
iex> schema = Zoi.time(coerce: true)
iex> Zoi.parse(schema, "12:34:56")
{:ok, ~T[12:34:56]}

  


        

      

      
        Transforms


        


  
    
      
    
    
      to_downcase(schema)



        
          
        

    

  


  

Converts a string to lowercase.
Example
iex> schema = Zoi.string() |> Zoi.to_downcase()
iex> Zoi.parse(schema, "Hello World")
{:ok, "hello world"}

  



  
    
      
    
    
      to_struct(schema, module)



        
          
        

    

  


  

      

          @spec to_struct(schema :: Zoi.Type.t(), struct :: module()) :: Zoi.Type.t()


      


Converts a schema to a struct of the given module.
This is useful for transforming parsed data into a specific struct type.
Example
defmodule User do
  defstruct [:name, :age]
end

schema = Zoi.object(%{
  name: Zoi.string(),
  age: Zoi.integer()
})
|> Zoi.to_struct(User)

Zoi.parse(schema, %{name: "Alice", age: 30})
#=> {:ok, %User{name: "Alice", age: 30}}

  



  
    
      
    
    
      to_upcase(schema)



        
          
        

    

  


  

      

          @spec to_upcase(schema :: Zoi.Type.t()) :: Zoi.Type.t()


      


Converts a string to uppercase.
Example
iex> schema = Zoi.string() |> Zoi.to_upcase()
iex> Zoi.parse(schema, "Hello World")
{:ok, "HELLO WORLD"}

  



  
    
      
    
    
      trim(schema)



        
          
        

    

  


  

      

          @spec trim(schema :: Zoi.Type.t()) :: Zoi.Type.t()


      


Trims whitespace from the beginning and end of a string.
Example
iex> schema = Zoi.string() |> Zoi.trim()
iex> Zoi.parse(schema, "  hello world  ")
{:ok, "hello world"}

  


        

      


  

    
Zoi.Context 
    



      
The Context provides the parsing information such as the input data, parsed data and errors.
The context is passed around during the parsing process to keep track of the current state of parsing.
It contains the schema being parsed, the input data, the parsed data, the path of the current error and any errors that have occurred during parsing.

        error()

      


    


    
      
        t()

      


    





  
    Functions
  


    
      
        add_error(context, error)

      


        Add a error to the context.

      error()



        
          
        

    

  


  

      

          @type error() :: Zoi.Error.t() | binary() | [Zoi.Error.t()]


      



  



  
    
      
    
    
      t()



        
          
        

    

  


  

      

          @type t() :: %Zoi.Context{
  errors: [Zoi.Error.t()],
  input: Zoi.input(),
  parsed: Zoi.input(),
  path: Zoi.Error.path(),
  schema: Zoi.Type.t()
}


      



  


        

      

      
        Functions


        


  
    
      
    
    
      add_error(context, error)



        
          
        

    

  


  

      

          @spec add_error(t(), error()) :: t()


      


Add a error to the context.
Example
iex> schema = Zoi.string() |> Zoi.refine(fn input, ctx ->
...>   if String.length(input) > 5 do
...>     :ok
...>   else
...>     Zoi.Context.add_error(ctx, %Zoi.Error{message: "Input too long"})
...>   end
...> end)
...> Zoi.parse(schema, "s")
{:error, [%Zoi.Error{message: "Input too long"}]}

  


        

      


  

    
Zoi.ISO 
    



      
This module defines schemas for ISO time, date, and datetime formats,
along with transformations to convert them into Elixir's native types.
It includes built-in transformations to convert ISO time, date, and datetime
strings into %Time{}, %Date{}, and %DateTime{} structs.

        date(opts \\ [])

      


        Defines a date type schema.



    


    
      
        datetime(opts \\ [])

      


        Defines a datetime type schema.



    


    
      
        naive_datetime(opts \\ [])

      


        Defines a naive datetime type schema.



    


    
      
        time(opts \\ [])

      


        Defines a time type schema.



    





  
    Transforms
  


    
      
        to_date_struct(schema)

      


        Converts Zoi.ISO.date() to %Date{} struct.



    


    
      
        to_datetime_struct(schema)

      


        Converts Zoi.ISO.datetime() to %DateTime{} struct.



    


    
      
        to_naive_datetime_struct(schema)

      


        Converts Zoi.ISO.naive_datetime() to %NaiveDateTime{} struct.

        Converts Zoi.ISO.time() to %Time{} struct.

      date(opts \\ [])



        
          
        

    

  


  

Defines a date type schema.
Example
iex> schema = Zoi.ISO.date()
iex> Zoi.parse(schema, "2025-08-07")
{:ok, "2025-08-07"}
iex> Zoi.parse(schema, "2025-02-30")
{:error, [%Zoi.Error{message: "invalid type: must be an ISO date"}]}

  



    

  
    
      
    
    
      datetime(opts \\ [])



        
          
        

    

  


  

Defines a datetime type schema.
Example
iex> schema = Zoi.ISO.datetime()
iex> Zoi.parse(schema, "2025-08-07T10:04:22+03:00")
{:ok, "2025-08-07T10:04:22+03:00"}

iex> schema = Zoi.ISO.datetime()
iex> Zoi.parse(schema, 1754646043)
{:error, [%Zoi.Error{message: "invalid type: must be an ISO datetime"}]}

  



    

  
    
      
    
    
      naive_datetime(opts \\ [])



        
          
        

    

  


  

Defines a naive datetime type schema.
Example
iex> schema = Zoi.ISO.naive_datetime()
iex> Zoi.parse(schema, "2025-08-07T10:04:22")
{:ok, "2025-08-07T10:04:22"}

iex> schema = Zoi.ISO.naive_datetime()
iex> Zoi.parse(schema, 1754646043)
{:error, [%Zoi.Error{message: "invalid type: must be an ISO naive datetime"}]}

  



    

  
    
      
    
    
      time(opts \\ [])



        
          
        

    

  


  

Defines a time type schema.
Example
iex> schema = Zoi.ISO.time()
iex> Zoi.parse(schema, "12:34:56")
{:ok, "12:34:56"}
iex> Zoi.parse(schema, "25:00:00")
{:error, [%Zoi.Error{message: "invalid type: must be an ISO time"}]}

      to_date_struct(schema)



        
          
        

    

  


  

Converts Zoi.ISO.date() to %Date{} struct.
Example
iex> schema = Zoi.ISO.date() |> Zoi.ISO.to_date_struct()
iex> Zoi.parse(schema, "2025-08-07")
{:ok, ~D[2025-08-07]}

  



  
    
      
    
    
      to_datetime_struct(schema)



        
          
        

    

  


  

Converts Zoi.ISO.datetime() to %DateTime{} struct.
Example
iex> schema = Zoi.ISO.datetime() |> Zoi.ISO.to_datetime_struct()
iex> Zoi.parse(schema, "2025-08-07T10:04:22+03:00")
{:ok, ~U[2025-08-07 07:04:22Z]}

  



  
    
      
    
    
      to_naive_datetime_struct(schema)



        
          
        

    

  


  

Converts Zoi.ISO.naive_datetime() to %NaiveDateTime{} struct.
Example
iex> schema = Zoi.ISO.naive_datetime() |> Zoi.ISO.to_naive_datetime_struct()
iex> Zoi.parse(schema, "2025-08-07T10:04:22")
{:ok, ~N[2025-08-07 10:04:22]}

  



  
    
      
    
    
      to_time_struct(schema)



        
          
        

    

  


  

Converts Zoi.ISO.time() to %Time{} struct.
Example
iex> schema = Zoi.ISO.time() |> Zoi.ISO.to_time_struct()
iex> Zoi.parse(schema, "12:34:56")
{:ok, ~T[12:34:56]}

  


        

      


  

    
Zoi.JSONSchema 
    



      
JSON Schema is a declarative language for annotating and validating JSON document's structure, constraints, and data types. It helps you standardize and define expectations for JSON data.
Zoi provides functionality to convert its type definitions into JSON Schema format, enabling seamless integration with systems that utilize JSON Schema for data validation and documentation.
Example
iex> schema = Zoi.object(%{name: Zoi.string(), age: Zoi.integer()})
iex> Zoi.to_json_schema(schema)
%{
  "$schema": "https://json-schema.org/draft/2020-12/schema",
  type: :object,
  properties: %{
    name: %{type: :string},
    age: %{type: :integer}
  },
  required: [:name, :age],
  additionalProperties: false
}
Supported Types
The following Zoi types are supported for conversion to JSON Schema:
	Zoi.string/0
	Zoi.integer/0
	Zoi.float/0
	Zoi.number/0
	Zoi.boolean/0
	Zoi.literal/1
	Zoi.null/0
	Zoi.array/1
	Zoi.tuple/1
	Zoi.enum/1
	Zoi.map/0
	Zoi.object/1
	Zoi.intersection/1
	Zoi.union/1
	Zoi.nullable/1
	Zoi.date/0 and Zoi.ISO.date/0
	Zoi.datetime/0 and Zoi.ISO.datetime/0
	Zoi.naive_datetime/0 and Zoi.ISO.naive_datetime/0
	Zoi.time/0 and Zoi.ISO.time/0

Metadata
Zoi.to_json_schema/1 will attempt to parse metadata defined on types. For more information, check the Zoi.metadata/1 documentation.
The following metadata will be parsed on JSON Schema conversion:
	:description - Adds a description field to the JSON Schema.
	:example - Adds an example field to the JSON Schema.

iex> schema = Zoi.string(metadata: [description: "A simple string", example: "Hello, World!"])
iex> Zoi.to_json_schema(schema)
%{
  "$schema": "https://json-schema.org/draft/2020-12/schema",
  type: :string,
  description: "A simple string",
  example: "Hello, World!"
}
Limitations
	Complex types or custom types not listed above will raise an error during conversion.
	Some advanced Zoi features may not have direct equivalents in JSON Schema.
	Refinements are partially supported, primarily for string patterns and length constraints.
	Additional properties in objects are disallowed by default (additionalProperties: false).

References
	JSON Schema Official Website

          @spec encode(Zoi.Type.t()) :: map()


      



  


        

      


  

    
Zoi.Struct 
    



      
A helper module to define and validate structs using Zoi schemas.
This module provides functions to extract @enforce_keys and struct fields from a Zoi struct schema.
It is particularly useful when you want to create Elixir structs that align with Zoi schemas.
Examples
defmodule MyApp.SomeModule do
  @schema Zoi.struct(__MODULE__, %{
    name: Zoi.string()
    age: Zoi.integer() |> Zoi.default(0) |> Zoi.optional(),
    email: Zoi.string()
  })

  @enforce_keys Zoi.Struct.enforce_keys(schema) # [:name]
  defstruct Zoi.Struct.struct_fields(schema) # [:name, :email, {:age, 0}]
end

    Functions
  


    
      
        enforce_keys(struct)

      


        Returns a list of keys that are required for the struct based on the schema.
This is useful for defining @enforce_keys in Elixir structs.



    


    
      
        struct_fields(struct)

      


        Returns a list of fields for the struct, where fields with default values are represented as tuples
of the form {key, default_value}.
This is useful for defining the fields of an Elixir struct.

Returns a list of keys that are required for the struct based on the schema.
This is useful for defining @enforce_keys in Elixir structs.

  



  
    
      
    
    
      struct_fields(struct)



        
          
        

    

  


  

Returns a list of fields for the struct, where fields with default values are represented as tuples
of the form {key, default_value}.
This is useful for defining the fields of an Elixir struct.

  


        

      


  

    
Zoi.Type protocol
    



      
Protocol for defining types in Zoi.
Types are used to validate and parse data according to a defined schema.
You can implement this protocol for custom types to handle specific validation and parsing logic.
Example
To create a custom type, you can define a module and implement the Zoi.Type protocol:
defmodule StringBoolean do
  use Zoi.Type.Def

  # `apply_type/1` is a helper function that will create the struct with the given options.
  def string_bool(opts \\ []) do
    apply_type(opts)
  end

  defimpl Zoi.Type do
    # This function is called to parse the input according to the schema.
    def parse(schema, input, opts) when is_binary(input) do
      {:ok, input}
    end

    def parse(schema, input, opts) when is_boolean(input) do
      {:ok, input}
    end

    def parse(_schema, _input, _opts) do
      {:error, "invalid string or boolean type"}
    end

    def type_spec(_schema, _opts) do
      quote do
        boolean() | binary()
      end
    end
  end
end
You can then use this type in your schema definitions and it will handle parsing and validation as defined.
iex> schema = StringBoolean.string_bool()
iex> Zoi.parse(schema, "hello world")
{:ok, "hello world"}
iex> Zoi.Type.parse(schema, true)
{:ok, true}
iex> Zoi.Type.parse(schema, 123)
{:error, %Zoi.Error{message: "invalid string or boolean type"}}

All the types that implement this protocol.

          @type t() :: %Zoi.Error{__exception__: true, message: binary(), path: path()}

